NAG Library Routine Document

F01QKF

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F01QKF returns the first ℓ rows of the real n by n orthogonal matrix P^{T}, where P is given as the product of Householder transformation matrices.

This routine is intended for use following F01QJF.

2 Specification

```
SUBROUTINE F01QKF (WHERET, M, N, NROWP, A, LDA, ZETA, WORK, IFAIL)
INTEGER             M, N, NROWP, LDA, IFAIL
REAL (KIND=nag_wp) A(LDA,*), ZETA(*), WORK(max(M-1,NROWP-M,1))
CHARACTER(1)        WHERET
```

3 Description

P is assumed to be given by

$$P = P_m P_{m-1} \cdots P_1$$

where

$$P_k = I - u_k u_k^{\mathrm{T}},$$

$$u_k = \begin{pmatrix} w_k \\ \zeta_k \\ 0 \\ z_k \end{pmatrix},$$

ζ_k is a scalar, w_k is a $(k-1)$ element vector and z_k is an $(n-m)$ element vector. w_k must be supplied in the kth row of A in elements $\mathrm{A}(k,1),\ldots,\mathrm{A}(k,k-1)$. z_k must be supplied in the kth row of A in elements $\mathrm{A}(k,m+1),\ldots,\mathrm{A}(k,n)$ and ζ_k must be supplied either in $\mathrm{A}(k,k)$ or in $\mathrm{ZETA}(k)$, depending upon the parameter WHERET.

4 References

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

Wilkinson J H (1965) *The Algebraic Eigenvalue Problem* Oxford University Press, Oxford

5 Parameters

1: WHERET – CHARACTER(1) *Input*

On entry: indicates where the elements of ζ are to be found.

WHERET = 'I' (In A)
The elements of ζ are in A.

WHERET = 'S' (Separate)
The elements of ζ are separate from A, in ZETA.

Constraint: WHERET = 'I' or 'S'.

2: M – INTEGER *Input*

On entry: m, the number of rows of the matrix A.

Constraint: $\mathrm{M} \geq 0$.

3: N – INTEGER *Input*

On entry: n, the number of columns of the matrix A.

Constraint: $\mathrm{N} \geq \mathrm{M}$.

4: NROWP – INTEGER *Input*

On entry: ℓ, the required number of rows of P.

If $\mathrm{NROWP} = 0$, an immediate return is effected.

Constraint: $0 \leq \mathrm{NROWP} \leq \mathrm{N}$.

5: A(LDA, $*$) – REAL (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array A must be at least $\max(1, \mathrm{N})$.

On entry: the leading m by m strictly lower triangular part of the array A, and the m by $(n - m)$ rectangular part of A with top left-hand corner at element $\mathrm{A}(1, \mathrm{M} + 1)$ must contain details of the matrix P. In addition, if $\mathrm{WHERET} = \text{'I'}$, the diagonal elements of A must contain the elements of ζ.

On exit: the first NROWP rows of the array A are overwritten by the first NROWP rows of the n by n orthogonal matrix P^{T}.

6: LDA – INTEGER *Input*

On entry: the first dimension of the array A as declared in the (sub)program from which F01QKF is called.

Constraint: $\mathrm{LDA} \geq \max(1, \mathrm{M}, \mathrm{NROWP})$.

7: ZETA($*$) – REAL (KIND=nag_wp) array *Input*

Note: the dimension of the array ZETA must be at least $\max(1, \mathrm{M})$ if $\mathrm{WHERET} = \text{'S'}$, and at least 1 otherwise.

On entry: with $\mathrm{WHERET} = \text{'S'}$, the array ZETA must contain the elements of ζ. If $\mathrm{ZETA}(k) = 0.0$ then P_k is assumed to be I, otherwise $\mathrm{ZETA}(k)$ is assumed to contain ζ_k.

When $\mathrm{WHERET} = \text{'I'}$, the array ZETA is not referenced.

8: WORK($\max(\mathrm{M} - 1, \mathrm{NROWP} - \mathrm{M}, 1)$) – REAL (KIND=nag_wp) array *Workspace*

Note: the dimension of the array WORK must be at least $\max(\mathrm{M} - 1, \mathrm{NROWP} - \mathrm{M}, 1)$.

9: IFAIL – INTEGER *Input/Output*

On entry: IFAIL must be set to 0, -1 or 1. If you are unfamiliar with this parameter you should refer to Section 3.3 in the Essential Introduction for details.

For environments where it might be inappropriate to halt program execution when an error is detected, the value -1 or 1 is recommended. If the output of error messages is undesirable, then the value 1 is recommended. Otherwise, if you are not familiar with this parameter, the recommended value is 0. **When the value -1 or 1 is used it is essential to test the value of IFAIL on exit.**

On exit: $\mathrm{IFAIL} = 0$ unless the routine detects an error or a warning has been flagged (see Section 6).

6 Error Indicators and Warnings

If on entry IFAIL $= 0$ or -1, explanatory error messages are output on the current error message unit (as defined by X04AAF).

Errors or warnings detected by the routine:

IFAIL $= -1$

On entry, WHERET $\neq$ 'I' or 'S',
or M < 0,
or N $<$ M,
or NROWP < 0 or NROWP $>$ N,
or LDA $<$ max(M, NROWP).

IFAIL $= -99$

An unexpected error has been triggered by this routine. Please contact NAG.

See Section 3.8 in the Essential Introduction for further information.

IFAIL $= -399$

Your licence key may have expired or may not have been installed correctly.

See Section 3.7 in the Essential Introduction for further information.

IFAIL $= -999$

Dynamic memory allocation failed.

See Section 3.6 in the Essential Introduction for further information.

7 Accuracy

The computed matrix P satisfies the relation

$$P = Q + E,$$

where Q is an exactly orthogonal matrix and

$$\|E\| \leq c\epsilon,$$

ϵ is the ***machine precision*** (see X02AJF), c is a modest function of n, and $\|.\|$ denotes the spectral (two) norm. See also Section 7 in F01QJF.

8 Parallelism and Performance

F01QKF is not threaded by NAG in any implementation.

F01QKF makes calls to BLAS and/or LAPACK routines, which may be threaded within the vendor library used by this implementation. Consult the documentation for the vendor library for further information.

Please consult the X06 Chapter Introduction for information on how to control and interrogate the OpenMP environment used within this routine. Please also consult the Users' Note for your implementation for any additional implementation-specific information.

9 Further Comments

The approximate number of floating-point operations is given by

$$\begin{aligned} &\tfrac{2}{3}m\{(3n-m)(2\ell-m) - m(\ell-m)\}, && \text{if } \ell \geq m, \text{ and} \\ &\tfrac{2}{3}\ell^2(3n-\ell), && \text{if } \ell < m. \end{aligned}$$

10 Example

This example obtains the 5 by 5 orthogonal matrix P following the RQ factorization of the 3 by 5 matrix A given by

$$A = \begin{pmatrix} 2.0 & 2.0 & 1.6 & 2.0 & 1.2 \\ 2.5 & 2.5 & -0.4 & -0.5 & -0.3 \\ 2.5 & 2.5 & 2.8 & 0.5 & -2.9 \end{pmatrix}.$$

10.1 Program Text

```
    Program f01qkfe

!     F01QKF Example Program Text

!     Mark 25 Release. NAG Copyright 2014.

!     .. Use Statements ..
      Use nag_library, Only: f01qjf, f01qkf, nag_wp
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, lda, ldpt, m, n, nrowp
!     .. Local Arrays ..
      Real (Kind=nag_wp), Allocatable  :: a(:,:), pt(:,:), work(:), zeta(:)
!     .. Executable Statements ..
      Write (nout,*) 'F01QKF Example Program Results'
      Write (nout,*)
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) m, n
      lda = m
      ldpt = n
      Allocate (a(lda,n),pt(ldpt,n),work(n),zeta(n))
      Read (nin,*)(a(i,1:n),i=1,m)

!     ifail: behaviour on error exit
!             =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
      ifail = 0
!     Find the RQ factorization of A
      Call f01qjf(m,n,a,lda,zeta,ifail)

!     Copy array A into PT and form the n by n matrix conjg(P')
      pt(1:m,1:n) = a(1:m,1:n)
      nrowp = n

      ifail = 0
      Call f01qkf('Separate',m,n,nrowp,pt,ldpt,zeta,work,ifail)

      Write (nout,*) 'Matrix  P'
      Write (nout,99999) pt(1:nrowp,1:n)


99999 Format (5(1X,F8.4))
    End Program f01qkfe
```

10.2 Program Data

```
F01QKF Example Program Data
  3     5                        : m, n
  2.0   2.0   1.6   2.0   1.2
  2.5   2.5  -0.4  -0.5  -0.3
  2.5   2.5   2.8   0.5  -2.9    : a
```

10.3 Program Results

```
 F01QKF Example Program Results

 Matrix  P
  -0.1310  -0.5170  -0.4642  -0.5054  -0.4946
  -0.1310  -0.5170  -0.4642   0.5054   0.4946
  -0.3276   0.5499  -0.5199  -0.3957   0.4043
  -0.6551   0.2494  -0.0928   0.4946  -0.5054
  -0.6551  -0.3175   0.5385  -0.2967   0.3032
```